

# ME 201/MTH 281/ME400/CHE400 Eigenfunction Expansion *Mathematica 7*

## ■ 1. Eigenfunctions and Eigenvalues

In this notebook, we consider a particular Sturm-Liouville system (the same as the one considered in section 4.2 of the class notes), and we determine the eigenvalues and eigenfunctions. Then as an example of an eigenfunction expansion, we expand a particular function in a series of the eigenfunctions. We check our expansion by a graphical comparison of the function and its series representation. The Sturm-Liouville system that is the basis for all this is given below. We use un-normalized eigenfunctions in this notebook.

$$\frac{d^2 y}{dx^2} + \lambda y = 0, 0 < x < L, \text{ with } y(0) = 0, \frac{dy}{dx}(L) = 0.$$

We solved this problem in class. We found the following eigenfunctions and eigenvalues:

$$\lambda[n\_ ] := (n - 1 / 2)^2 \pi^2 / L^2; \mathbf{y}[x\_ , n\_ ] := \mathbf{Sin}[(n - 1 / 2) * \pi * x / L]$$

According to the general theory, these eigenfunctions are orthogonal on  $[0, L]$ . We verify that.

```
int = Integrate[y[x, n] y[x, m], {x, 0, L}]
```

$$\frac{L \left( \frac{\sin[(m-n) \pi]}{m-n} - \frac{\sin[(-1+m+n) \pi]}{-1+m+n} \right)}{2 \pi}$$

Now we tell *Mathematica* that  $m$  and  $n$  are integers. We do this by using the Simplify command, adding the restriction that  $m$  and  $n$  are contained in the set Integers.

```
Simplify[int, {m, n} ∈ Integers]
```

```
0
```

This verifies the orthogonality. The normalization integral corresponds to the case  $m = n$ .

```
normint = Integrate[y[x, n] y[x, n], {x, 0, L}]
```

$$\frac{1}{2} \left( L + \frac{L \sin[2 n \pi]}{(-1 + 2 n) \pi} \right)$$

```
norm[n_] = Simplify[normint, n ∈ Integers]
```

$$\frac{L}{2}$$

## ■ 2. An Eigenfunction Expansion

To illustrate the construction of an eigenfunction expansion, we expand the function

```
f[x_] := x * (L - x)
```

We will use the eigenfunctions  $y[x,n]$ , which are not normalized. We first calculate the expansion coefficients.

```
c[n_] =
  Simplify[Integrate[y[x, n] f[x], {x, 0, L}], n ∈ Integers] / norm[n]
```

$$\frac{8 L^2 (4 + (-1)^n (-1 + 2 n) \pi)}{(-1 + 2 n)^3 \pi^3}$$

Now we define the  $k$ th partial sum of the eigenfunction series as  $fsum[x,k]$ .

```
fsum[x_, k_] := Sum[c[n] * y[x, n], {n, 1, k}]
```


Now we make a sequence of 21 graphs, showing in the  $k$ th graph a plot of  $f[x]$  in blue and of  $fsum[x,k]$  in red. We first set  $L$  equal to 1. When the graphs are animated, they give a dynamic view of the convergence.

```
L = 1;
```

```
graph[k_] := Plot[{f[x], fsum[x, k]}, {x, 0, L},
  AxesLabel → {"x", "f"}, PlotRange → {0, 0.3}, ImageSize → 250,
  PlotStyle → {{RGBColor[0, 0, 1], Thickness[0.004]},
    {RGBColor[1, 0, 0], Thickness[0.004]}},
  PlotLabel → Row[{"k =", PaddedForm[k, 2]}]]
```



We construct the sequence of 21 graphs and group them into a single cell. To play the sequence as a movie, select the group and then go to the Menu item **Graphics->Rendering->Animate Selected Graphics**.




```
Do[Print[graph[k]], {k, 1, 21}];
```



For the printed version of this notebook, we show every 5th graph of this sequence.

```
Do[Print[graph[k]], {k, 1, 21, 5}];
```





We see that the convergence is generally quite good, except very near  $x = 1$ . At  $x = 1$ , the eigenfunctions have zero derivative whereas the function we are expanding does not. This creates the extended tail in the series partial sum near  $x = 1$ .

What is the rate of convergence of this series? It is easy to determine by looking at the  $n$ th term, which is given by

**c[n] y[x, n]**

$$\frac{8 \left(4 + (-1)^n (-1 + 2n) \pi\right) \text{Sin}\left[\left(-\frac{1}{2} + n\right) \pi x\right]}{(-1 + 2n)^3 \pi^3}$$

We see that for large  $n$ , the terms drop off like  $1/n^2$  which is not too bad. In general we would need on the order of 10 terms for 1% or smaller error.